title: Kotlin build FatFramework with defined minimum iOS version
body: <p>I want to deploy my multiplatform project to .framework with supported iOS architecture using FatFramework using this gradle configuration. It's work but I founded in info.plist that it have default MinimumOSVersion to iOS 9 so I can't used my library below that version.</p>
  
  <p>Is there any way / possible to configure the minimum iOS version for my lib ?  </p>
  
  <pre><code>import org.jetbrains.kotlin.gradle.tasks.FatFrameworkTask
  
      plugins {
          id("org.jetbrains.kotlin.multiplatform") version "1.3.61"
          id("maven-publish")
          id("com.jfrog.bintray") version "1.8.4"
          id("org.jetbrains.dokka") version "0.10.0"
      }
  
      group = "com.example.lib"
      version = "1.0.0"
  
  
      repositories {
          mavenCentral()
          jcenter()
      }
  
  
      kotlin {
          jvm()
          val iosX64 = iosX64("ios")
          val iosArm64 = iosArm64("iosArm64")
          val iosArm32 = iosArm32("iosArm32")
  
          val frameworkName = "EXAMPLE-LIB"
  
          configure(listOf(iosX64, iosArm64, iosArm32)) {
              binaries.framework {
                  baseName = frameworkName
              }
          }
  
          sourceSets {
              named("commonMain") {
                  dependencies {
                      implementation(kotlin("stdlib-common"))
                      implementation(kotlin("stdlib-jdk7"))
                      implementation(kotlin("stdlib-jdk8"))
                      implementation("org.jetbrains.kotlinx:kotlinx-coroutines-core-common:1.3.0-RC")
                      implementation(kotlin("reflect"))
                  }
              }
              named("commonTest") {
                  dependencies {
                      implementation(kotlin("test-common"))
                      implementation(kotlin("test-annotations-common"))
                      implementation(kotlin("reflect"))
                  }
              }
              named("jvmMain") {
                  dependencies {
                      implementation(kotlin("stdlib-jdk8"))
                  }
              }
              named("jvmTest") {
                  dependencies {
                      implementation(kotlin("test"))
                      implementation(kotlin("test-junit"))
                  }
              }
              val iosMain by getting {
                  dependencies {
                      implementation("org.jetbrains.kotlinx:kotlinx-coroutines-core-native:1.3.0-RC")
                      implementation("io.ktor:ktor-client-ios:1.2.6")
                  }
              }
              val iosArm64Main by getting {
                  dependencies {
                      implementation("org.jetbrains.kotlinx:kotlinx-coroutines-core-native:1.3.0-RC")
                      implementation("io.ktor:ktor-client-ios:1.2.6")
                  }
              }
              val iosArm32Main by getting {
                  dependencies {
                      implementation("org.jetbrains.kotlinx:kotlinx-coroutines-core-native:1.3.0-RC")
                      implementation("io.ktor:ktor-client-ios:1.2.6")
                  }
              }
          }
  
          val debugFatFramework by tasks.creating(FatFrameworkTask::class) {
              baseName = frameworkName
              from(
                  iosArm32.binaries.getFramework("debug"),
                  iosArm64.binaries.getFramework("debug"),
                  iosX64.binaries.getFramework("debug")
              )
              destinationDir = buildDir.resolve("fat-framework/debug")
              group = "Universal framework"
              description = "Builds a debug universal (fat) framework"
          }
  
          val releaseFatFramework by tasks.creating(FatFrameworkTask::class) {
              baseName = frameworkName
              from(
                  iosArm32.binaries.getFramework("release"),
                  iosArm64.binaries.getFramework("release"),
                  iosX64.binaries.getFramework("release")
              )
              destinationDir = buildDir.resolve("fat-framework/release")
              group = "Universal framework"
              description = "Builds a release universal (fat) framework"
          }
  
          val zipDebugFatFramework by tasks.creating(Zip::class) {
              dependsOn(debugFatFramework)
              from(debugFatFramework)
              from("LICENSE.md")
          }
  
          val zipReleaseFatFramework by tasks.creating(Zip::class) {
              dependsOn(releaseFatFramework)
              from(releaseFatFramework)
              from("LICENSE.md")
          }
  
  
      }
  
      tasks.register&lt;Exec&gt;("generateThriftModels") {
          executable = "java"
          args(
              "-jar", "thrifty-compiler.jar",
              "--lang", "kotlin",
              "--omit-generated-annotations",
              "--list-type=kotlin.collections.ArrayList",
              "--set-type=kotlin.collections.LinkedHashSet",
              "--out", "src/commonMain/kotlin",
              "--path", "./thrift/",
              "./thrift/nativeapp.thrift"
          )
      }
  </code></pre>